

ГЕНЕРАТОР НА ОСНОВЕ КОНТРОЛЛЕРА ARDUINO И СИНТЕЗАТОРА AD9835, ПРЕДНАЗНАЧЕННЫЙ ДЛЯ НАСТРОЙКИ ФИЛЬТРОВ С РЕЗОНАТОРАМИ

Кулаков Владимир Геннадьевич
SPIN РИНЦ: 2111-7702

Контакт с автором: kulakovvlge@gmail.com

Данная статья продолжает тему об использовании синтезатора частоты для создания перестраиваемого генератора синусоидального сигнала, предназначенного для настройки фильтров с кварцевыми и керамическими резонаторами. В статье рассматривается генератор на основе контроллера Arduino и синтезатора AD9835.

Любые ошибки и неточности, допущенные в процессе расчета или технической реализации фильтра, содержащего в своем составе кварцевые резонаторы, могут привести к тому, что данный фильтр просто не пропустит полезный сигнал, так как его резонансная частота будет смещена относительно несущей частоты сигнала. Проблема заключается в том, что у фильтров с резонаторами полоса пропускания очень узкая – она может составлять менее одной десятитысячной доли от резонансной частоты фильтра. Далеко не всякий лабораторный генератор сигналов может обеспечить необходимую для решения данной задачи плавность перестройки частоты.

Цифровой синтезатор AD9835, выпускаемый фирмой Analog Devices, способен формировать сигналы синусоидальной формы с частотой от 0 до 25 МГц. Частота и фаза выходного сигнала AD9835 являются программируемыми: программирование осуществляется с внешнего микроконтроллера по интерфейсу SPI, причем можно использовать широко распространенные контроллеры линейки Arduino.

При работе на высоких частотах начинает проявляться один из недостатков цифровых синтезаторов – становится очень заметной ступенчатость выходного сигнала. Однако полосовые фильтры, построенные на основе резонаторов, сами по себе обладают способностью подавлять помехи, поэтому для проверки их параметров можно использовать и не слишком качественные сигналы, имеющие значительный уровень шумов. Следовательно, можно использовать очень простую конструкцию генератора.

На рисунке 1 в качестве примера показана принципиальная схема генератора, построенного на основе контроллера Arduino, микросхемы AD9835BRUZ и инкрементального энкодера. Частота сигнала на выходе генератора определяется информацией, поступающей с датчика на контроллер Arduino, программой, загруженной в этот контроллер, а также частотой тактового сигнала, подаваемого на микросхему синтезатора AD9835 с кварцевого генератора DD1.

Напряжение питания +5В, необходимое для работы кварцевого генератора DD1 и синтезатора DA1, обеспечивает стабилизатор напряжения, встроенный в контроллер Arduino.




Рисунок 1. Высокочастотный генератор на микросхеме AD9835

В данном примере используется тактовый сигнал с частотой 50 МГц. Код K выходной частоты F для синтезатора AD9835 в таком случае должен вычисляться по следующей формуле:

$$K = F \times 2^{32} / 50000000.$$

Результат вычислений должен быть округлен до 32-разрядного целого числа без знака, а затем разделен на четыре части и побайтно загружен в регистр синтезатора, хранящий код частоты.

Пример программы, позволяющей перестраивать частоту генератора с шагом 50 Гц в диапазоне от 11,99 МГц до 12,01 МГц, приведен в листинге 1.

Листинг 1. Программа для управления работой синтезатора AD9835

```
#include <SPI.h>
#define pin_CLK 2 // Энкодер - пин А
#define pin_DT 4 // Энкодер - пин В
#define PosLim 200 // Ограничитель изменения положения

const float fgen = 5.0E7; // частота сигнала ГТИ, Гц
const float sf = 12.0E6; // начальное значение выходной частоты, Гц
const float df = 50.0; // шаг приращения частоты, Гц

volatile long Pos = 0; // текущее значение положения энкодера
long oldPos = 0; // предыдущее положение энкодера
float ff; // код частоты в формате с плавающей точкой
unsigned long f; // двоичный код выходной частоты

// Составные части кода частоты для загрузки в регистр FREQ0
unsigned int hmsbs, lmsbs, hlsbs, llsbs;

// Процедура для обработки прерывания от энкодера
void EncoderRotate() {
    if (digitalRead(pin_CLK) == digitalRead(pin_DT)) {
        if(Pos < PosLim) Pos++;
    } else {
        if(Pos > -PosLim) Pos--;
    }
}

// Процедура для передачи данных синтезатору AD9835
// через интерфейс SPI
void WriteAD9835(uint16_t Data){
    SPI.beginTransaction(SPISettings(SPI_CLOCK_DIV2,
    MSBFIRST, SPI_MODE2));
    digitalWrite(SS, LOW);
    delayMicroseconds(1);
    SPI.transfer16(Data);
    digitalWrite(SS, HIGH);
    SPI.endTransaction();
}

void setup() {
    pinMode(pin_CLK, INPUT);
    pinMode(pin_DT, INPUT);
    attachInterrupt(digitalPinToInterrupt(pin_CLK),
    EncoderRotate, RISING);
    // Вычисляем код частоты в формате с плавающей точкой
    ff = sf*(float)0x100001*(float)0x100001/fgen;
    // Получаем двоичный код выходной частоты и делим его на части
    f = (unsigned long)ff;
    llsbs = ((unsigned int)f & 0xff) | 0x2000;
    f >>= 8;
    hlsbs = ((unsigned int)f & 0xff) | 0x3100;
    f >>= 8;
    lmsbs = ((unsigned int)f & 0xff) | 0x2200;
```

```

f >>= 8;
hmsbs = ((unsigned int)f & 0xff) | 0x3300;

// Запускаем интерфейс SPI
SPI.begin();
// Сбрасываем генератор в исходное состояние
WriteAD9835(0xF800);
// Загружаем код частоты
WriteAD9835(hmsbs);
WriteAD9835(lmsbs);
WriteAD9835(hlsbs);
WriteAD9835(llsbs);
// Запускаем генератор
WriteAD9835(0xC000);
}

void loop() {
  if (oldPos != Pos)
  {
    // Запомнить текущую позицию
    oldPos = Pos;
    // Вычисляем код частоты в формате с плавающей точкой
    ff = ((sf + df*(float)Pos)*(float)0x100001*(float)0x100001)/fgen;
    // Получаем двоичный код выходной частоты и делим его на части
    f = (unsigned long)ff;
    llsbs = ((unsigned int)f & 0xff) | 0x2000;
    f >>= 8;
    hlsbs = ((unsigned int)f & 0xff) | 0x3100;
    f >>= 8;
    lmsbs = ((unsigned int)f & 0xff) | 0x2200;
    f >>= 8;
    hmsbs = ((unsigned int)f & 0xff) | 0x3300;
    // Загружаем новый код частоты в синтезатор
    WriteAD9835(hmsbs);
    WriteAD9835(lmsbs);
    WriteAD9835(hlsbs);
    WriteAD9835(llsbs);
    // Перезапускаем генератор
    WriteAD9835(0x9000);
  }
}

```

При повороте ручки энкодера в ту или иную сторону генерируется прерывание и контроллер либо увеличивает, либо уменьшает значение переменной-счетчика положения энкодера Pos. После любого изменения значения данной переменной контроллер заново вычисляет код частоты и загружает его в микросхему AD9835.

Форма сигнала на выходе синтезатора AD9835 при частоте колебаний 12 МГц показана на рисунке 2.




Рисунок 2. Осциллограмма сигнала на выходе AD9835 при частоте 12 МГц

Подключим к выходу генератора Т-образный фильтр с попарно включенными резонаторами, имеющими номинальную частоту 12 МГц. Схема фильтра приведена на рисунке 3.




Рисунок 3. Схема Т-образного фильтра с попарно включенными резонаторами

Осциллограмма сигнала на выходе фильтра на частоте резонанса (12,0018 МГц) показана на рисунке 4.




Рисунок 4. Осциллограмма сигнала на выходе фильтра на частоте резонанса

Список использованной литературы

1. 50 MHz Direct Digital Synthesizer, Waveform Generator AD9835 – Analog Devices, Inc., 2011.
2. AN-621 application note. Programming the AD9832/AD9835 by Colm Slattery – Analog Devices, Inc., 2003.
3. AN-1108 application note. AD9832/AD9835 Programming Examples by Liam Riordan – Analog Devices, Inc., 2011.
4. Кулаков В.Г. Применение керамических резонаторов для преобразования меандра в синусоидальный сигнал. [Электронный ресурс]. URL: <http://new-idea.kulichki.net/pubfiles/200621074046.pdf> (дата обращения: 21.06.2020).
5. Кулаков В.Г. Определение ширины полосы пропускания фильтров на кварцевых и керамических резонаторах при помощи генератора на микросхеме AD9833. [Электронный ресурс]. URL: <http://new-idea.kulichki.net/pubfiles/200808081704.pdf> (дата обращения: 08.08.2020).
6. Кулаков В.Г. Генератор на микросхеме AD9833 с инкрементальным энкодером. [Электронный ресурс]. URL: <http://new-idea.kulichki.net/pubfiles/200810082331.pdf> (дата обращения: 10.08.2020).
7. Кулаков В.Г. Высокочастотный генератор на микросхеме AD9833 с многозвенным RC-фильтром. [Электронный ресурс]. URL: <http://new-idea.kulichki.net/pubfiles/200820074529.pdf> (дата обращения: 20.08.2020).
8. Кулаков В.Г. Т-образный фильтр с кварцевыми резонаторами, соединенными попарно. [Электронный ресурс]. URL: <http://new-idea.kulichki.net/pubfiles/210208175339.pdf> (дата обращения: 08.02.2021).